
A1-BROUWER DEGREES IN MACAULAY2

NIKITA BORISOV, THOMAS BRAZELTON, FRENLY ESPINO, THOMAS HAGEDORN,
ZHAOBO HAN, JORDY LOPEZ GARCIA, JOEL LOUWSMA, WERN JUIN GABRIEL ONG,

AND ANDREW R. TAWFEEK

Abstract. We describe the Macaulay2 package A1BrouwerDegrees for computing local
and global A1-Brouwer degrees and studying symmetric bilinear forms over the complex
numbers, the real numbers, the rational numbers, and finite fields of characteristic not
equal to 2.

1. Introduction

In A1-homotopy theory, the A1-Brouwer degree provides an algebro-geometric analogue of
the classical Brouwer degree from differential topology. Morel’s A1-degree homomorphism
identifies the zeroth stable stem of the motivic sphere spectrum with the Grothendieck–Witt

ring of symmetric bilinear forms over a field [18, Corollary 1.24]. Given an endomorphism
of affine space with an isolated rational zero, work of Kass and Wickelgren [10] identifies
its local A1-Brouwer degree with the Eisenbud–Khimshiashvili–Levine signature form [5, 7],
which was used to compute local Brouwer degrees in real differential topology. Work of
Bachmann and Wickelgren [2] extends this work, identifying the A1-Brouwer degree with a
quadratic Grothendieck–Serre duality form.

In A1-enumerative geometry [11, 15] (see [3, 20] for an overview), the A1-Brouwer degree
has found a wealth of applications, recently including [1, 9, 8]. For instance, via McK-
ean’s Bézout theorem, the A1-Brouwer degree can be understood as a quadratically enriched
analogue of intersection multiplicity, often encoding deeper geometric information than was
available over the algebraically closed fields – with other invariants of the quadratic form
over k capturing field-specific arithmetic data [16].

Recent work of the second-named author, McKean, and Pauli [4] provides tractable formu-
las for computing A1-Brouwer degrees as Bézoutian bilinear forms. In the A1BrouwerDegrees
package, we implement these methods in Macaulay2 [6] over the fields C, R, Q, and Fq (for
q odd) and provide a suite of tools whose capabilities include:

(1) computing A1-Brouwer degrees (both local and global) for endomorphisms of affine
space;1

(2) decomposing symmetric bilinear forms into their isotropic and anisotropic parts; and
(3) extracting invariants of symmetric bilinear forms (rank, signature, discriminant,

Hasse–Witt invariants).

Remark 1. The current scope of this package is the complex numbers, the real numbers, the
rational numbers, and finite fields of characteristic not equal to 2, so all of our algorithms are

2020 Mathematics Subject Classification. 14F42, 14-04, 68W30, 11E04, 55M25, 14N10.
1Due to R being an inexact field, A1-Brouwer degrees over R have to be computed over Q and base-changed

to R.

implicitly taking place in these settings. We hope to expand the scope of these algorithms
in future work.

In Section 2, we provide a rapid introduction to the theory of symmetric bilinear forms,
highlighting the capacity of our package to build forms, check isomorphisms, and decompose
forms. In Section 3, we discuss local and global A1-Brouwer degrees and provide some
computational examples, including quadratically enriched intersection multiplicity of real
curves, the A1-Euler characteristic of the Grassmannian Gr(2, 4) (following [4, Section 8.2]),
and local computations for 27 lines on a cubic surface (following [11, 19]).

1.1. Software availability. The software documented here is available in versions 1.23 and
later of Macaulay2 as the A1BrouwerDegrees package.

2. The Grothendieck–Witt ring

For this entire section, we assume k is a field of characteristic not equal to 2. We say a
bilinear form β : V × V → k is symmetric if β(v, w) = β(w, v) for all v, w ∈ V . We say β is
non-degenerate if β(v,−) : V → k is identically zero if and only if v = 0.

Definition 2. Let β : V ×V → k be a symmetric bilinear form, and choose a basis e1, . . . , en
for V . We define the Gram matrix of β in the basis {ei}

n
i=1 to be the symmetric matrix with

entries β(ei, ej).

Remark 3. Non-degeneracy of β is equivalent to the statement that the determinant of a
Gram matrix in any basis is nonzero. A change of basis for V corresponds to the associated
Gram matrices being congruent.

Given two symmetric bilinear forms βi : Vi × Vi → k for i = 1, 2, we can define their sum
and product:

(1)
(β1 ⊕ β2) : (V1 ⊕ V2)× (V1 ⊕ V2)→ k

(β1 ⊗ β2) : (V1 ⊗ V2)× (V1 ⊗ V2)→ k.

On Gram matrices, these operations are given by direct sum and tensor product, respectively.

Definition 4. The Grothendieck–Witt ring GW(k) is the group completion of the semiring
of isomorphism classes of non-degenerate symmetric bilinear forms over k.

Example 5. Any non-degenerate symmetric bilinear form over an algebraically closed field
admits a basis in which its Gram matrix is the identity; therefore rank determines an iso-
morphism GW(C) ∼= Z. For further computations of Grothendieck–Witt rings, we refer the
reader to [14, Chapter II].

When the field k is the complex numbers, the real numbers, the rational numbers, or a
finite field (of characteristic not 2), we define a type called GrothendieckWittClass that
encodes the class [β] ∈ GW(k) of a symmetric bilinear form β. Grothendieck–Witt classes
can be constructed from Gram matrices via the makeGWClass method.

i1 : needsPackage "A1BrouwerDegrees";

i2 : M = matrix(QQ, {{1,3},{3,7}});

o2 : Matrix Q2 ←− Q2

i3 : beta = makeGWClass M

https://macaulay2.com/doc/Macaulay2/share/doc/Macaulay2/A1BrouwerDegrees/html/index.html

o3 =

(
1 3
3 7

)

o3 : GrothendieckWittClass

Given a Grothendieck–Witt class beta, its underlying field can be obtained by run-
ning getBaseField beta, and the underlying matrix can be obtained by running either
beta.matrix or getMatrix beta. Objects of type GrothendieckWittClass can be added
and multiplied via the addGW and multiplyGW methods, respectively.

Example 6. For any unit a ∈ k×, there is a symmetric bilinear form of rank one

〈a〉 : k × k → k

(x, y) 7→ axy.

Via the change of basis (x, y) 7→ (bx, by) for any unit b ∈ k×, we observe that 〈a〉 = 〈ab2〉.
Hence the representative for a rank one form is determined only by its square class.

The following classical result (see [14, Corollary I.2.4]) implies that the forms 〈a〉 generate
GW(k).

Theorem 7. Every class in GW(k) is represented by a diagonal Gram matrix.

A diagonal representative of a class in GW(k) can be found using the getDiagonalClass
method.

i4 : getDiagonalClass beta

o4 =

(
1 0
0 −2

)

o4 : GrothendieckWittClass

We also provide methods for constructing various forms. We can construct a class cor-
responding to a list of diagonal entries via the makeDiagonalForm method. We denote by
〈a1, . . . , an〉 the direct sum of the rank one forms 〈ai〉 for 1 ≤ i ≤ n.

i5 : makeDiagonalForm(GF(13), (2,6))

o5 =

(
2 0
0 6

)

o5 : GrothendieckWittClass

Hyperbolic forms are crucial objects of study due to their local-to-global behavior (see
Theorem 9), and they can be produced via the makeHyperbolicForm method. Similarly,
Pfister forms, which are important objects of study in the world of quadratic forms [14,
Chapter X], can be produced via the makePfisterForm method.

2.1. Verifying isomorphisms of forms. Given two non-degenerate symmetric bilinear
forms, a natural question is whether they represent the same element of GW(k). An easy
invariant to check is whether they are defined on vector spaces of the same dimension, i.e.,
whether the rank of the forms (the rank of their Gram matrices) agrees. As mentioned in
Example 5, since C is algebraically closed and every number is a square, rank completely
classifies symmetric bilinear forms over the complex numbers.

Since there are two square classes over the real numbers, namely +1 and −1, we can find a
Gram matrix representative of any form which is diagonal, with only ±1 appearing along the

diagonal. The trace of such a Gram matrix is an invariant of the form, called the signature.
Rank and signature jointly classify symmetric bilinear forms over the real numbers.

i6 : gamma = makeGWClass matrix(RR, {{3,0,0},{0,-4,0},{0,0,7}});

i7 : getSignature gamma

o7 = 1

Over finite fields, the discriminant, which is the determinant of any Gram matrix repre-
sentative (valued in square classes), and the rank jointly classify symmetric bilinear forms.

Over the rational numbers, the classification of symmetric bilinear forms is more com-
plicated. The isomorphism class of a non-degenerate form β is determined by its rank,
discriminant dβ ∈ Q∗/(Q∗)2, signature (as a form over R), and Hasse–Witt invariants at all
finite primes p. For a prime p, the Hasse–Witt invariant [17, III5, p.79] is defined as follows.

Definition 8. Given any form β ∼= 〈a1, . . . , an〉 ∈ GW(Q), its Hasse–Witt invariant εp(β)
at a prime p is the product ∏

i<j

(ai, aj)p,

where (−,−)p denotes the Hilbert symbol

(a, b)p :=

{
1 if z2 = ax2 + by2 has a nonzero solution in Qp,

−1 otherwise.

We can compute the Hilbert symbol (a, b)p via getHilbertSymbol(a,b,p) and the Hasse–
Witt invariant of a form beta at p by getHasseWittInvariant(beta, p). If p is an odd
prime and the p-adic valuations νp(a) and νp(b) are even, then (a, b)p = 1. Thus, εp(β) is 1
for almost all primes p and only needs to be computed for p = 2 and odd primes p with
νp(dβ) odd.

These methods together form one of our core methods isIsomorphicForm, which is a
Boolean-valued method that determines whether two symmetric bilinear forms are isomor-
phic. This is done by reference to the relevant invariants over C, R, Fq (for q odd), or Q.

2.2. Decomposing forms. Witt’s Decomposition Theorem (see [14, I.4.1]) implies that
any non-degenerate symmetric bilinear form decomposes into an anisotropic part and an
isotropic part that is a sum of hyperbolic forms. This decomposition is crucial in simplifying
an element of GW(k). While this decomposition is fairly routine over C, R, and Fq, to
decompose forms over Q we must implement existing algorithms from the literature. An
important mathematical stepping stone is the following local-to-global principle for isotropy,
a reference for which is [14, VI.3.1].

Theorem 9 (Hasse–Minkowski Principle). A form β ∈ GW(Q) is isotropic if and only if it

is isotropic over R and over Qp for all primes p.

Our method getAnisotropicDimensionQQp, an implementation of [12, Algorithm 8], de-
termines the dimension of the anisotropic part of a symmetric bilinear form over Qp. The
method getAnisotropicDimension returns the anisotropic dimension of a form defined over
the real numbers, the complex numbers, a finite field, or the rational numbers.

Given a form, we can therefore decompose it as

β ∼= βa ⊕ nH,

where βa is anisotropic, H denotes the hyperbolic form 〈1,−1〉, and n is the Witt index

(implemented as getWittIndex).
The Boolean-valued method isAnisotropic returns whether a form is anisotropic; the

method isIsotropic is its negation.

i8 : alpha = makeDiagonalForm(QQ, (1,2,-3));

i9 : isAnisotropic alpha

o9 = false

i10 : isIsotropic alpha

o10 = true

Over Q, the computation of the anisotropic part of β is carried out inductively by reduction
of the anisotropic dimension of β, following recently published algorithms for quadratic forms
over number fields by Koprowski and Rothkegel [13]. The anisotropic part of a form can be
computed via getAnisotropicPart.

i11 : beta = makeDiagonalForm(QQ, (3,-3,2,5,1,-9));

i12 : getAnisotropicPart beta

o12 =

(
2 0
0 5

)

o12 : GrothendieckWittClass

A quick string reading off the decomposition of a form can be obtained by running the
getSumDecompositionString method.

i13 : getSumDecompositionString beta

o13 = 2H + <2> + <5>

3. A1-Brouwer degrees

For the symbolic computations in this section, let k be an exact field2 of characteristic not
equal to 2. The methods for computing A1-Brouwer degrees only work for polynomials with
isolated zeros [4, Theorem 1.2].

In [4], the authors show that the local and global A1-Brouwer degrees of an endomorphism
of affine space with isolated zeros can be expressed in terms of a bilinear form associated to
the Bézoutian of the endomorphism.

More explicitly, for fi ∈ k[x1, . . . , xn], suppose f = (f1, . . . , fn) : A
n
k → An

k has isolated ze-
ros. Introducing new variables (X1, . . . , Xn) and (Y1, . . . , Yn), we can construct the matrix ∆
with entries

∆i,j =
fi(Y1, . . . , Yj−1, Xj , . . . , Xn)− fi(Y1, . . . , Yj, Xj+1, . . . , Xn)

Xj − Yj

.

One can think of this matrix ∆ as a Jacobian of formal derivatives. Define Q(f) =
k[x1, . . . , xn]/(f1, . . . , fn) and Qp(f) = k[x1, . . . , xn]m/(f1, . . . , fn) for m the maximal ideal
of a closed point p in the preimage of 0. The Bézoutian of f is defined to be the image of

2An exact field is a field whose elements are represented exactly by Macaulay2, e.g., Q or Fq.



6 A1
-BROUWER DEGREES IN MACAULAY2

det(∆) in the algebra Q(f)⊗Q(f) (respectively, in the local algebra Qp(f)⊗Qp(f)). Given
a1, . . . , am a k-linear basis of Q(f) (resp., Qp(f)), there are bi,j ∈ k such that

det(∆) =
∑

1≤i≤j≤m

bi,j(ai ⊗ aj)

in Q(f) ⊗ Q(f) (resp., Qp(f)⊗ Qp(f)). The Bézoutian bilinear form, the symmetric bilin-
ear form with Gram matrix given by the bi,j, gives the global (resp., local) A1-degree [4,
Theorem 1.2].

In the case of the global A1-degree, a theorem of Macaulay tells us that a k-basis of the
algebra Q(f) is given by the standard monomials (see [21, Proposition 2.1]). In the case
of the local A1-degree, a k-basis for the local ring can be calculated via the quotient of
k[x1, . . . , xn] by the saturation of I = (f1, . . . , fn) at m.

Proposition 10 ([21, Proposition 2.5]). The natural map xi 7→ xi defines an isomorphism

of rings

(2) k[x1, . . . , xn]m/I ∼= k[x1, . . . , xn]/(I : (I : m∞)),

where I is an ideal of k[x1, . . . , xn] and (I : (I : m∞)) is the quotient of I by the saturation

of I at m.

The getLocalAlgebraBasis(I, m)method uses this isomorphism to find a basis ofQp(f).
It determines a k-basis of the right side of Equation (2) as a k-vector space. Proposition 10
then gives a k-basis of Qp(f).

These methods for computing k-bases for Q(f) and Qp(f) allow us to algorithmically
implement techniques to compute the global and local A1-degrees (see also [4, Section 5A]).

3.1. A univariate polynomial. A univariate polynomial over a field k defines an endo-
morphism of affine space A1

k → A1
k. Consider the endomorphism f : A1

Q → A1
Q defined

by
f(x) = (x2 + x+ 1)(x− 3)(x+ 2).

We can compute the global degree.

i14 : R = QQ[x];

i15 : f = {x^4 - 6*x^2 - 7*x - 6};

i16 : alpha = getGlobalA1Degree f

o16 =




−7 −6 0 1
−6 0 1 0
0 1 0 0
1 0 0 0




o16 : GrothendieckWittClass

We can also compute the local degrees at the ideals (x2 + x + 1), (x − 3), and (x + 2),
respectively.

i17 : I1 = ideal(x^2 + x + 1);

o17 : Ideal of R

i18 : alpha1 = getLocalA1Degree(f, I1)

o18 =

(
−5 −7
−7 −2

)

o18 : GrothendieckWittClass

i19 : I2 = ideal(x - 3)

o19 = ideal (x− 3)

o19 : Ideal of R
i20 : alpha2 = getLocalA1Degree(f, I2)

o20 =
(
65

)

o20 : GrothendieckWittClass

i21 : I3 = ideal(x + 2);

o21 : Ideal of R
i22 : alpha3 = getLocalA1Degree(f, I3)

o22 =
(
−15

)

o22 : GrothendieckWittClass

We can then use the isIsomorphicForm method (see also Section 2.1) to verify that the
local A1-degrees sum to the global A1-degree.

i23 : alpha’ = addGW(alpha1,addGW(alpha2,alpha3));

i24 : isIsomorphicForm(alpha,alpha’)

o24 = true

Consider the graph of f(x).

−3 −2 −1 1 2 3

−20

20

40

60

Following [16, Theorem 1.2], A1-degrees can be understood as enriched intersection numbers,
determined by the signed volume of the parallelepiped spanned by the gradient vectors of
the hypersurfaces at the intersection point. In the one-dimensional case, considering the
normal vectors, we can interpret α2 = 〈65〉, the local A1-degree at (x− 3), and α3 = 〈−15〉,
the local A1-degree at (x+ 2), as signs of the derivative at these points.

3.2. The Euler characteristic of the Grassmannian of lines in P3. For k a field
of characteristic not 2, let Grk(2, 4) be the Grassmannian of lines in P3

k. Following [4,
Example 8.2], we can compute the A1-Euler characteristic of the Grassmannian over k = F27

as the A1-degree of the section σ : A4
F27
→ A4

F27
defined by3

(x1, x2, x3, x4) 7→ (x2 − x1x3, 1− x1x4, x4 − x1 − x2

3,−x2 − x3x4).

3There is a small error in the definition of σ in [4, Example 8.2]. The second and third component
functions of σ should be swapped in order to agree with the ordered basis induced on the tangent bundle of
Grk(2, 4) as in [11, Proposition 45]. By [4, Example 6.3], the overall computation is only affected by a sign.

We compute the A1-Euler characteristic as follows.

i25 : k = GF(27);

i26 : x = symbol x;

i27 : R = k[x_1,x_2,x_3,x_4];

i28 : f = {x_2 - x_1*x_3, 1 - x_1*x_4, x_4 - x_1 - x_3^2, -x_2 - x_3*x_4};

i29 : beta = getGlobalA1Degree f

o29 =




0 0 0 0 0 1
0 1 0 0 0 0
0 0 0 −1 0 0
0 0 −1 0 0 0
0 0 0 0 1 0
1 0 0 0 0 0




o29 : GrothendieckWittClass

We can subsequently use the getSumDecompositionString method to decompose the sym-
metric bilinear form β.

i30 : getSumDecompositionString beta

o30 = 2H + <1> + <1>

Our computation agrees with the result given in [4, Example 8.2] and shows

χ(GrF27
(2, 4)) = 2H+ 〈1〉+ 〈1〉.

3.3. Local geometry of some lines on the Fermat cubic surface. In their pioneering
paper [11], Kass and Wickelgren give a Grothendieck–Witt class-valued count of the number
of lines on a smooth cubic surface, providing an interpretation of the local A1-degree as the
topological type of the line. To illustrate some features of the A1BrouwerDegrees package,
we use it to compute the topological type of some lines on the Fermat cubic surface.

Let k be a field, and let {e1, e2, e3, e4} be the standard basis for k4. By [11, Lemma 45],
we can define local coordinates on Spec(k[y1, y2, y3, y4]) ∼= A4

k around the point of Grk(2, 4)
defined by the span of {e3, e4} such that y1, y2, y3, y4 corresponds to the span of {ẽ3, ẽ4},
where

ẽi =






ei for i ∈ {1, 2},

e1y1 + e2y2 + e3 for i = 3,

e1y3 + e2y4 + e4 for i = 4.

Letting S denote the tautological bundle over Grk(2, 4), the above coordinates provide a
trivialization of the vector bundle Sym3 S∨ over the open affine subvariety

U ∼= Spec(k[y1, y2, y3, y4]) ⊆ Grk(2, 4).

A cubic surface X defines a section σX |U : U → Sym3 S∨|U that vanishes on the lines on X
that, when treated as affine two-dimensional subspaces of k4, contain e3 and e4 in their span.

Let us consider the Fermat cubic surface defined by the homogeneous cubic equation
x3
0 + x3

1 + x3
2 + x3

3. That is,

X =
{
[x0 : x1 : x2 : x3] ∈ P3

k : x
3

0 + x3

1 + x3

2 + x3

3 = 0
}
⊆ P3

k.

Working over Q, the lines on X are all defined over the cyclotomic extension Q(ζ) for ζ a
primitive third root of unity. We can explicitly compute the 27 lines as

[s : t : −ζ it : −ζjs], [s : t : −ζ is : −ζjt], [s : −ζ is : t : −ζjt]

for 0 ≤ i, j ≤ 2 and [s : t] ∈ P1
Q. Note that there are only 18 lines containing e3 and e4 in

their span. We thus expect the section to vanish at 18 points. Applying Pauli’s computation
of sections of σX |U : U → Sym3 S∨|U in [19, §2.2, Remark 2.7], our section is of the form
σX |U = (f1, f2, f3, f4) where

f1(y1, y2, y3, y4) = y31 + y33 + 1

f2(y1, y2, y3, y4) = 3y21y2 + 3y23y4

f3(y1, y2, y3, y4) = 3y1y
2

2 + 3y3y
2

4

f4(y1, y2, y3, y4) = y32 + y34 + 1.

We compute the global A1-degree, which is rank 18, as expected.

i31 : R = QQ[y_1,y_2,y_3,y_4];

i32 : f = {y_1^3 + y_3^3 + 1,

3*y_1^2*y_2 + 3*y_3^2*y_4,

3*y_1*y_2^2 + 3*y_3*y_4^2,

y_2^3 + y_4^3 + 1};

i33 : alpha = getGlobalA1Degree f;

i34 : getSumDecompositionString alpha

o34 = 8H + <1> + <1>

To compute the local degree, we find an isolated zero using the minimalPrimes method of
Macaulay2.

i35 : I = (minimalPrimes ideal f)_0

o35 = ideal (y4, y3 + 1, y2 + 1, y1)

o35 : Ideal of R

We then compute the local A1-degree at this point.

i36 : beta = getLocalA1Degree(f, I)

o36 =
(
81

)

o36 : GrothendieckWittClass

i37 : getSumDecomposition beta

o37 =
(
1
)

o37 : GrothendieckWittClass

At the other nine minimal primes, the same calculation gives the local degrees as one copy
of 〈1〉, six copies of 〈3,−1〉 and two copies of 〈2,−6〉. As

6〈3,−1〉+ 2〈2,−6〉 ∼= 8H,

the global A1-degree equals the sum of the local A1-degrees.
The local computation above indicates that the line spanned by {−e2 + e3,−e1 + e4} on

the Fermat cubic surface is a hyperbolic line. We briefly show this agrees with the type as
defined in [11, Definition 9].

By [11, Proposition 14], the local type of the line is equal to the resultant of the par-
tial derivatives of the equation of the Fermat cubic surface restricted to the line. Letting
z1, z2, z3, z4 be the dual basis to e1, e2, e3, e4 defined above, we can write the equation of
the Fermat surface in terms of the dual basis via the change of basis z1 7→ z1 + z4, z2 7→
z2 + z3, z3 7→ −z3, z4 7→ −z4 so that the line is spanned by e3 and e4.

i38 : needsPackage "Resultants";

i39 : R = QQ[z_1,z_2][z_3,z_4];

i40 : fermat = (z_1 + z_4)^3 + (z_2 + z_3)^3 - z_3^3 - z_4^3;

We compute the restriction of the partial derivatives of the defining equation of the Fermat
cubic surface to the surface with respect to the dual basis z1, z2.

i41 : g1 = sub(diff(z_1, fermat), {z_1 => 0, z_2 => 0});

i42 : g2 = sub(diff(z_2, fermat), {z_1 => 0, z_2 => 0});

We then compute the resultant of these polynomials and consider it as a quadratic form
over Q in order to agree with the computation of the local index over Q.

i43 : line_type = makeDiagonalForm(QQ, lift(resultant {g1,g2}, QQ))

o43 =
(
81

)

o43 : GrothendieckWittClass

i44 : isIsomorphicForm(line_type,beta)

o44 = true

Thus this computation agrees with the local A1-degree of the associated section of Sym3 S∗

as computed above.

